Programs for Recursive Identification

Söderström, Torsten

1973

Document Version:
Publisher's PDF, also known as Version of record

Link to publication

Citation for published version (APA):
Söderström, T. (1973). *Programs for Recursive Identification*. (Technical Reports TFRT-7049). Department of Automatic Control, Lund Institute of Technology (LTH).

Total number of authors:
1

PROGRAMS FOR RECURSIVE IDENTIFICATION

T. SÖDERSTRÖM

Report 7312 (C) July 1973
Lund Institute of Technology
Division of Automatic Control

PROGRAMS FOR RECURSIVE IDENTIFICATION.

T. Söderström

ABSTRACT.

In this report some programs for recursive identification are described and listed.

TABLE OF CONTENTS

		Page
I.	Description of the Programs	1
II.	References	5
III.	Lists of the Programs	
IV.	Examples	

I. DESCRIPTION OF THE PROGRAMS.

The described program package uses the subroutines SIMUL, NODI, PRB, PRBSTA, NSTAB, SLOSS, RTLS (described in the program library of the Division of Automatic Control), PMPY (described in Söderström (1973 b)), MERRO, VML, MVS and PADD (described in Söderström (1973 c)).

1. TRTID. This is a test program, which generates data, manages the administration and prints the results.

2. General description of the subroutines
   RTGLS, RTIVM1, RTIVM2 and RTMLE. The routines perform one iteration of the actual recursive identification method. They must be called once for each sampling interval. The initial values of the parameter estimates must be given in the calling program at the first call. It is assumed generally that all signals for negative times are zero. The initial value of the estimated covariance matrix (or the corresponding matrix) is put 100 · I in the routines. This means that the user is urged to scale all signals to an amplitude of roughly 1.0.

3. RTGLS. The model used is

$$\hat{A}(q^{-1})y(t) = \hat{B}(q^{-1})u(t) + \frac{1}{\hat{C}(q^{-1})}\varepsilon(t) \tag{1}$$

$$\hat{A}(q^{-1}) = 1 + \hat{a}_1 q^{-1} + \ldots + \hat{a}_{NA} q^{-NA}$$

$$\hat{B}(q^{-1}) = \hat{b}_1 q^{-1} + \ldots + \hat{b}_{NB} q^{-NB}$$

$$\hat{C}(q^{-1}) = 1 + \hat{c}_1 q^{-1} + \ldots + \hat{c}_{NC} q^{-NC}$$

The algorithm is described in Harting-James-Sage (1969).

Special parameters.

JF - determines if the data are to be filtered or not,

JC - determines if the C-parameters are to be estimated or not.

Rule of thumb. Put JF = 0 for the first 20 - 50 samples and then put JF = 1. Put JC = 1 the whole time.

4. RTIVM1 and RTIVM2. The model used is

$$\hat{A}(q^{-1})y(t) = \hat{B}(q^{-1})u(t) + \varepsilon(t) \tag{2}$$

$$\hat{A}(q^{-1}) = 1 + \hat{a}_1 q^{-1} + \ldots + \hat{a}_{NA} q^{-NA}$$

$$\hat{B}(q^{-1}) = \hat{b}_1 q^{-1} + \ldots + \hat{b}_{NB} q^{-NB}$$

The basic algorithm is described in Wong-Polak (1967) and Young (1970).

RTIVM1: The instrumental variables are chosen as old values of the input signal.

RTIVM2: The instrumental variables are chosen as the input signal and the estimated deterministic output signal.

Special parameter.

ND - the estimated deterministic output is computed using the parameter estimates obtained ND sampling intervals ago.

Rule of thumb. Choose ND roughly = NA. It is wise to try some different values.

5. RTMLE. The model used is

$$\hat{A}(q^{-1})y(t) = \hat{B}(q^{-1})u(t) + \hat{C}(q^{-1})\varepsilon(t)$$

$$\hat{A}(q^{-1}) = 1 + \hat{a}_1 q^{-1} + \dots + \hat{a}_{NA} q^{-NA}$$

$$\hat{B}(q^{-1}) = \hat{b}_1 q^{-1} + \dots + \hat{b}_{NB} q^{-NB}$$

$$\hat{C}(q^{-1}) = 1 + \hat{c}_1 q^{-1} + \dots + \hat{c}_{NC} q^{-NC}$$

The algorithm is described in Söderström (1973 a).

An estimated covariance matrix of the parameters estimates can easily be computed by

P(I,J) = V*V TTIN(I,J)*2./FLOAT(ND)

where ND is the actual number of sampling intervals.

It is described in Söderström (1973 a) how to use restarts and processing the computations without updating the estimates (i.e. to put INEW = 1) in order to improve the algorithm. Compare also the construction in TRTID.

<u>Rule of thumb</u>. The variables described in Söderström (1973 a) may be chosen as

N1 = 100 - 300

N2 = 25 - 100

$$VTEST = 1 + (2-3)\,\frac{NA + NB + NC}{N1}$$

The subroutine VML evaluates the asymptotic normalized loss function, see Söderström (1973 c).

NB. It is necessary that $z^{NC} + \hat{c}_1 z^{NC-1} + \ldots + \hat{c}_{NC}$ has all zeros inside the unit circle for the initial values.

II. REFERENCES.

Hasting-James, R. - Sage, M.W. (1969).
Recursive Generalized-Least-Squares Procedure for On-Line Identification of Process Parameters. Proc. IEE, Vol. 116, No. 12, pp 2057 - 2062.

Söderström, T. (1973 a).
An On-Line Algorithm for Approximate Maximum Likelihood Identification of Linear Dynamic Systems. Report 7308, Division of Automatic Control, Lund Institute of Technology.

Söderström, T. (1973 b).
Computation of Asymptotic Least Squares Estimates. Report 7314(C), Division of Automatic Control, Lund Institute of Technology.

Söderström, T. (1973 c).
Programs for Evaluation of Identified Models of Simulated Data. Report 7315(C), Division of Automatic Control, Lund Institute of Technology.

Wong, K.Y. - Polak, E. (1967).
Identification of Linear Discrete Time Systems Using the Instrumental Variable Method. IEEE Trans. Aut. Ctrl., Vol. AC-12, No. 6, pp 707 - 718.

Young, P.C. (1970).
An Extension of the Instrumental Variable Method for Identification of a Noisy Dynamic Process. Univ. of Cambridge, Dep. of Eng., Technical note CN/70/1.

III. LISTS OF THE PROGRAMS

```
      PROGRAM TRTID
      TEST OF REAL TIME IDENTIFICATION METHODS
      AUTHOR TORSTEN SÖDERSTRÖM 1973-04-12

      SUBROUTINE REQUIRED
         SIMUL
            PRBSTA
            PRB
            NODI
         RTGLS
         RTIVM1
         RTIVM2
         RTMLE
         VML
         RTLS

      DIMENSION U(1000),Y(1000),TS(30),TM(30),P(30,30)
      DIMENSION F(20)
      DIMENSION A1(11),B1(11),C1(11),T1(30)

    1 READ 100,IMETH,NSA,NSB,NSC,NMA,NMB,NMC,M
  100 FORMAT(16I5)
      IF(IMETH.EQ.0) STOP
      NS=NSA+NSB+NSC
      READ 101,(TS(I),I=1,NS),AL
  101 FORMAT(8F10.5)
      IF(IMETH.EQ.1) READ 100,NC
      IF(IMETH.EQ.3) READ 100,ND
      IF(IMETH.EQ.4) READ 100,N1,N2
      NM=NMA+NMB+NMC
      DO 5 I=1,NM
      T1(I)=0.
    5 TM(I)=0.
      VTEST=1.+2.*FLOAT(NM)/N1
      ICONV=0
      VE=AL**2
      IPRINT=25

      PRINT 200
  200 FORMAT(1H1,10X,'REAL TIME IDENTIFICATION')
      IF(IMETH.EQ.1) PRINT 201
      IF(IMETH.EQ.2) PRINT 202
      IF(IMETH.EQ.3) PRINT 203
      IF(IMETH.EQ.4) PRINT 204
      IF(IMETH.EQ.5) PRINT 208
  201 FORMAT(/10X,'RTGLS IS USED')
  202 FORMAT(/10X,'RTIVM1 IS USED')
  203 FORMAT(/10X,'RTIVM2 IS USED')
  204 FORMAT(/10X,'RTMLE IS USED')
  208 FORMAT(/10X,'RTLS IS USED')
      PRINT 205
  205 FORMAT(/10X,'TRUE PARAMETER VALUES')
      PRINT 101,(TS(I),I=1,NS),AL
      PRINT 206
  206 FORMAT(/10X,'NSA NSB NSC NMA NMB NMC M')
      PRINT 100,NSA,NSB,NSC,NMA,NMB,NMC,M
      IF(IMETH.EQ.1) PRINT 100,NC
      IF(IMETH.EQ.3) PRINT 100,ND
      IF(IMETH.EQ.4) PRINT 100,N1,N2
      PRINT 207
  207 FORMAT(/10X,'TIME AND PARAMETER ESTIMATES')

      GENERATION OF DATA
```

```
C
      IE=19
      CALL SIMUL(U,Y,TS,1.0,AL,M,NSA,NSB,NSC,1,7,IE)
C
C     IDENTIFICATION LOOP
C
      K1=MSLEFT(DUM)
      DO 70 K=1,M
      UU=U(K)
      YY=Y(K)
      INIT=0
      IF(K.EQ.1) INIT=1
C
C     CHOOSE METHOD
C
      GO TO (10,20,30,40,50),IMETH
C
C
   10 JC=0
      IF(K.GT.NC) JC=1
      CALL RTGLS(TM,P,YY,UU,RES,NMA,NMB,NMC,30,1,JC,INIT)
      GO TO 60
C
C
   20 CALL RTIVM1(TM,UU,YY,RES,NMA,NMB,INIT)
      GO TO 60
C
C
   30 CALL RTIVM2(TM,UU,YY,RES,NMA,NMA,ND,INIT)
      GO TO 60
C
C
   40 INEW=0
      IF(ICONV.EQ.1) GO TO 41
      MM=MOD(K,N1+N2)
      IF(MM.EQ.(N1+1)) INIT=1
      IF(MM.GE.(N1+1)) INEW=1
   41 CALL RTMLE(TM,V,P,UU,YY,RES,NMA,NMB,NMC,INIT,INEW,30)
      IF(ICONV.EQ.1) GO TO 60
      IF(MM.NE.N1) GO TO 42
      VEEST=V/(MM)
      VU=1.0/VEEST
      CALL VML(T1,TM,NSA,NSB,NSC,NMA,NMB,NMC,VU,1.0,IS,W)
      DO 43 I=1,NM
   43 T1(I)=TM(I)
      IF(W.LE.VTEST) ICONV=1
      PRINT 209,W,VTEST
  209 FORMAT(10X,'W AND VTEST',2G12.5)
   42 CONTINUE
      GO TO 60
C
C
   50 IF(INIT.NE.1) GO TO 52
      DO 51 I=1,NM
      DO 511 J=1,NM
  511 P(I,J)=0.
      P(I,I)=100.
   51 F(I)=0.
   52 CALL RTLS(TM,P,F,YY,NM,30,1.,RES)
      IF(NMA-1) 56,55,53
   53 DO 54 I=NMA,2,-1
   54 F(I)=F(I-1)
   55 F(1)=-YY
   56 IF(NMB-1) 599,59,57
```

```
 57 DO 58 I=NMB,2,-1
 58 F(NMA+I)=F(NMA+I-1)
 59 F(NMA+1)=UU
599 CONTINUE
C
 60 IF(MOD(K,IPRINT).EQ.0) PRINT 300,K,(TM(I),I=1,NM),RES
300 FORMAT(I5,5X,10G12.5)
 70 CONTINUE
C
C     EVALUATION OF MODELS
C
    CALL MERRO(TS,TM,NSA,NSB,NMA,NMB,W1)
    IF(IMETH.EQ.1) GO TO 71
    CALL VML(TS,TM,NSA,NSB,NSC,NMA,NMB,NMC,1.0,VE,IS,W2)
    CALL MVS(TS,TM,NSA,NSB,NSC,NMA,NMB,NMC,VE,IS,W3)
    GO TO 90
 71 A1(1)=1.
    C1(1)=1.
    IF(NMA.EQ.0) GO TO 73
    DO 72 I=1,NMA
 72 A1(I+1)=TM(I)
 73 IF(NMB.EQ.0) GO TO 75
    DO 74 I=1,NMB
 74 B1(I)=TM(NMA+I)
 75 IF(NMC.EQ.0) GO TO 77
    DO 76 I=1,NMC
 76 C1(I+1)=TM(NMA+NMB+I)
 77 NA1=NMA+1
    NC1=NMC+1
    CALL PMPY(T1,NNA,A1,NA1,C1,NC1)
    NNA=NNA-1
    IF(NNA.EQ.0) GO TO 79
    DO 78 I=1,NNA
 78 T1(I)=T1(I+1)
 79 CALL PMPY(F,NNB,B1,NMB,C1,NC1)
    IF(NNB.EQ.0) GO TO 81
    DO 80 I=1,NNB
 80 T1(NNA+I)=F(I)
 81 NN=NNA+NNB
    CALL VML(TS,T1,NSA,NSB,NSC,NNA,NNB,0,1.0,VE,IS,W2)
    CALL MVS(TS,T1,NSA,NSB,NSC,NNA,NNB,0,VE,IS,W3)
C
 90 PRINT 210,W1
210 FORMAT(//10X,'VARIANCE OF THE MODEL ERROR',G12.5)
    PRINT 211,W2
211 FORMAT(10X,'VARIANCE OF THE RESIDUALS',G12.5)
    PRINT 212,W3
212 FORMAT(10X,'VARIANCE OF THE OUTPUT',G12.5)
C
    K2=MSLEFT(DUM)
    K=K1-K2
    PRINT 400,K
400 FORMAT(//10X,'TIME FOR THIS IDENTIFICATION WAS',I5,' MS')
    GO TO 1
    END
```

```
      SUBROUTINE RTGLS(T,P,Y,U,RES,NA,NB,NC,IB,JF,JC,INIT)
C
C     PERFORMS  ONE ITERATION OF THE RECURSIVE GLS METHOD
C     THE FOLLOWING MODEL IS USED
C     A(Q)*C(Q)*Y(T)=B(Q)*C(Q)*U(T)+RES(T)
C     A(Q)=1+A(1)*Q**(-1)+...+A(NA)*Q**(-NA)
C     B(Q)=  B(1)*Q**(-1)+...+B(NB)*Q**(-NB)
C     C(Q)=1+C(1)*Q**(-1)+...+C(NC)*Q**(-NC)
C     E(T) WHITE NOISE
C
C     AUTHOR TORSTEN SÖDERSTRÖM 1972-06-24
C     REVISED TORSTEN SÖDERSTRÖM 1973-04-10
C     REFERENCE REPORT 7312(C)
C
C     T-VECTOR OF ORDER (NA+NB+NC) CONTAINING THE PARAMETER ESTIMATES
C       T=( A(1)...C(NC) )
C     P-MATRIX OF ORDER (NA+NB+NC) * (NA+NB+NC)
C       CONTAINING AN ESTIMATED VARAIANCE MATRIX OF T
C     Y=Y(T)
C     U=U(T)
C     RES=RES(T)
C     NA,NB,NC ORDER OF THE MODEL (MIN 0,MAX 10)
C     IB DIMENSION PARAMETER
C     JF=0 NO FILTERING
C     JF=1    FILTERING
C     JC=1 C-PARAMETERS ARE ESTIMATED
C     JC=0 C-PARAMETERS ARE NOT ESTIMATED
C     INIT -IF INIT=1 P AND INTERNAL VARIABLES ARE GIVEN STARTVALUES
C                     P=100. * THE UNIT MATRIX
C                     IT IS ASSUMED THAT ALL OLD U(T) AND Y(T) ARE 0.
C         ELSE THE ALGORITHM IS APPLIED STRAIGHTFORWARD
C
C     SUBROUTINE REQUIRED
C            RTLS
C
      DIMENSION T(1),P(IB,1)
      DIMENSION F1(40),F2(20),F3(10),T1(10),P1(10,10)
C
      N2=NA+NB
      N3=N2+NC
      N4=NA+NB+NC*2
      IF(N3.EQ.0) RETURN
C
C     INITIAL VALUES
C
      IF(INIT.NE.1) GO TO 10
      IF(NC.EQ.0) GO TO 2
      DO 1 I=1,NC
    1 F3(I)=0.
    2 IF(N2.EQ.0) GO TO 5
      DO 3 I=1,N2
    3 F2(I)=0.
      DO 4 I=1,N4
    4 F1(I)=0.
    5 DO 7 I=1,N3
      DO 6 J=1,N3
    6 P(I,J)=0.
    7 P(I,I)=100.
   10 CONTINUE
C
C     FILTER DATA
C
      YF=Y
```

```
      IF(NA.EQ.0) GO TO 12
      DO 11 I=1,NA
   11 F2(I)=F1(I)
   12 IF(NB.EQ.0) GO TO 14
      DO 13 I=1,NB
   13 F2(NA+I)=F1(NA+NC+I)
   14 CONTINUE
      IF(JF.EQ.0) GO TO 20
      IF(NC.EQ.0) GO TO 20
      DO 15 I=1,NC
   15 YF=YF-T(N2+I)*F1(I)
      IF(NA.EQ.0) GO TO 17
      DO 16 I=1,NA
      DO 16 J=1,NC
   16 F2(I)=F2(I)+T(N2+J)*F1(I+J)
   17 IF(NB.EQ.0) GO TO 20
      DO 18 I=1,NB
      DO 18 J=1,NC
   18 F2(NA+I)=F2(NA+I)+T(N2+J)*F1(NA+NC+I+J)
   20 CONTINUE

      ESTIMATE A AND B PARAMETERS

      CALL RTLS(T,P,F2,YF,N2,IB,1.,RES)


      COMPUTE RESIDUALS AND ESTIMATE THE C-PARAMETERS

      IF(JC.EQ.0) GO TO 30
      IF(NC.EQ.0) GO TO 30
      E=Y
      IF(NA.EQ.0) GO TO 22
      DO 21 I=1,NA
   21 E=E-T(I)*F1(I)
   22 IF(NB.EQ.0) GO TO 24
      DO 23 I=1,NB
   23 E=E-T(NA+I)*F1(NA+NC+I)
   24 DO 25 I=1,NC
      T1(I)=T(N2+I)
      DO 25 J=1,NC
   25 P1(I,J)=P(N2+I,N2+J)

      CALL RTLS(T1,P1,F3,E,NC,10,1.,RES)

      DO 26 I=1,NC
      T(N2+I)=T1(I)
      DO 26 J=1,NC
   26 P(N2+I,N2+J)=P1(I,J)
   30 CONTINUE

      UPDATE DATA VECTORS

      NN=NA+NC
      IF(NN-1) 34,33,31
   31 DO 32 I=NN,2,-1
   32 F1(I)=F1(I-1)
   33 F1(1)=-Y
   34 NN=NA+NB
      IF(NN-1) 38,37,35
   35 DO 36 I=NN,2,-1
   36 F1(NA+NC+I)=F1(NA+NC+I-1)
   37 F1(NA+NC+1)=U
   38 CONTINUE
```

```
      IF(JC.EQ.0) GO TO 50
      IF(NC-1) 50,41,39
   39 DO 40 I=NC,2,-1
   40 F3(I)=F3(I-1)
   41 F3(1)=-E
   50 CONTINUE
C
      RETURN
      END
```

```
      SUBROUTINE RTIVM1(T,U,Y,RES,NA,NB,INIT)
C
C     PERFORMS ONE ITERATION OF THE RECURSIVE INSTRUMENTAL VARIABLE METHOD
C     VERSION 1
C     THE INPUT IS USED AS INSTRUMENTAL VARIABLES
C
C     THE FOLLOWING MODEL IS USED
C     A(Q)*Y(T)=B(Q)*U(T)+RES(T)
C     A(Q)=1+A(1)*Q**(-1)+...A(NA)*Q**(-NA)
C     B(Q)=  B(1)*Q**(-1)+...B(NB)*Q**(-NB)
C     V(T) INDEPENDENT OF U(T). THE SPECTRUM OF V(T) MAY BE ARBIBTRARY
C
C     AUTHOR TORSTEN SÖDERSTRÖM 1970-12-24
C     REVISED TORSTEN SÖDERSTRÖM 1973-04-12
C     REFERENCE REPORT 7312(C)
C
C     T-VECTOR OF ORDER (NA+NB) CONTAINING TNE PARAMETER ESTIMATES
C       T=(A(1)...A(NA) B(1)...B(NB))
C     U=U(T)
C     Y=Y(T)
C     RES=RES(T)
C     NA,NB MODEL ORDER (MIN 0,MAX 10)
C       (NA+NB).GT.0
C     INIT-IF INIT=1 INTERNAL VARAIBLES ARE GIVEN STARTVALUES
C                    IT IS ASSUMED THAT OLD U(T) AND Y(T) ARE ZERO
C                    IT IS ASSUMED THAT U(T) AND Y(T) ARE OF MAGNITUDE 1.0
C        ELSE THE ALGORITHM IS APPLIED STRAIGHTFORWARD
C
C     SUBROUTINE REQUIRED
C             NONE
C
      DIMENSION T(1)
      DIMENSION P(20,20),F(20),Z(20),H1(20),H2(20)
C
      N=NA+NB
      IF(INIT.NE.1) GO TO 10
      DO 2 I=1,N
      F(I)=0.
      Z(I)=0.
      DO 1 J=1,N
    1 P(I,J)=0.
    2 P(I,I)=100.
C
   10 DO 11 I=1,N
      H1(I)=0.
      H2(I)=0.
      DO 11 J=1,N
      H1(I)=H1(I)+P(J,I)*F(J)
   11 H2(I)=H2(I)+P(I,J)*Z(J)
      A=1.
      RES=Y
      DO 12 I=1,N
      A=A+H1(I)*Z(I)
   12 RES=RES-F(I)*T(I)
      DO 13 I=1,N
   13 H2(I)=H2(I)/A
C
      DO 14 I=1,N
      T(I)=T(I)+H2(I)*RES
      DO 14 J=1,N
   14 P(I,J)=P(I,J)-H2(I)*H1(J)
C
C     UPDATE F AND Z
```

```
C
      IF(N.EQ.1) GO TO 22
      DO 21 I=N,2,-1
   21 Z(I)=Z(I-1)
   22 Z(1)=U
      IF(NA-1) 26,25,23
   23 DO 24 I=NA,2,-1
   24 F(I)=F(I-1)
   25 F(1)=-Y
   26 IF(NB-1) 30,29,27
   27 DO 28 I=NB,2,-1
   28 F(NA+I)=F(NA+I-1)
   29 F(NA+1)=U
   30 CONTINUE
C
      RETURN
      END
```

```
      SUBROUTINE RTIVM2(T,U,Y,RES,NA,NB,ND,INIT)
C
C     PERFORMS ONE ITERATION OF THE RECURSIVE INSTRUMENTAL VARIABLE METHOD
C     VERSION 2
C     THE INPUT AND THE ESTIMATED DETERMINISTIC OUTPUT ARE USED AS
C     INSTRUMENTAL VARIABLES
C
C     THE FOLLOWING MODEL IS USED
C     A(Q)*Y(T)=B(Q)*U(T)+RES(T)
C     A(Q)=1+A(1)*Q**(-1)+...A(NA)*Q**(-NA)
C     B(Q)=  B(1)*Q**(-1)+...B(NB)*Q**(-NB)
C     V(T) INDEPENDENT OF U(T). THE SPECTRUM OF V(T) MAY BE ARRIBTRARY
C
C     AUTHOR TORSTEN SÖDERSTRÖM 1970-12-24
C     REVISED TORSTEN SÖDERSTRÖM 1973-04-12
C     REFERENCE REPORT 7312(C)
C
C     T-VECTOR OF ORDER (NA+NB) CONTAINING TNE PARAMETER ESTIMATES
C       T=(A(1)...A(NA) B(1)...B(NB))
C     U=U(T)
C     Y=Y(T)
C     RES=RES(T)
C     NA,NB MODEL ORDER (MIN 0,MAX 10)
C       (NA+NB).GT.0
C     ND -THE MODEL IS DELAYED ND STEPS BEFORE IT IS USED FOR COMPUTATION
C        OF THE ESTIMATED DETERMINISTIC OUTPUT (MIN 0,MAX 10)
C     INIT-IF INIT=1 INTERNAL VARAIBLES ARE GIVEN STARTVALUES
C                    IT IS ASSUMED THAT OLD U(T) AND Y(T) ARE ZERO
C                    IT IS ASSUMED THAT U(T) AND Y(T) ARE OF MAGNITUDE 1.0
C        ELSE THE ALGORITHM IS APPLIED STRAIGHTFORWARD
C     SUBROUTINE REQUIRED
C        NSTAB
C
      DIMENSION T(1)
      DIMENSION P(20,20),F(20),Z(20),H1(20),H2(20),D(20,10)
C
      N=NA+NB
      IF(INIT.NE.1) GO TO 10
      DO 2 I=1,N
      F(I)=0.
      Z(I)=0.
      DO 1 J=1,N
    1 P(I,J)=0.
      M=MAX0(1,ND)
      DO 3 J=1,M
    3 D(I,J)=0.
    2 P(I,I)=100.
C
   10 DO 11 I=1,N
      H1(I)=0.
      H2(I)=0.
      DO 11 J=1,N
      H1(I)=H1(I)+P(J,I)*F(J)
   11 H2(I)=H2(I)+P(I,J)*Z(J)
      A=1.
      RES=Y
      DO 12 I=1,N
      A=A+H1(I)*Z(I)
   12 RES=RES-F(I)*T(I)
      DO 13 I=1,N
   13 H2(I)=H2(I)/A
C
      DO 14 I=1,N
```

```
      T(I)=T(I)+H2(I)*RES
      DO 14 J=1,N
   14 P(I,J)=P(I,J)-H2(I)*H1(J)
C
C     COMPUTE YEST AND UPDATE D
C
      IF(ND.GT.0) GO TO 32
      YEST=0.
      DO 31 I=1,N
   31 YEST=YEST-T(I)*F(I)
      GO TO 38
   32 DO 33 I=1,N
   33 YEST=YEST-D(I,1)*F(I)
C
      IF(NA.EQ.0) GO TO 34
      IF(NSTAB(T,NA).EQ.-1) GO TO 38
   34 IF(ND.EQ.1) GO TO 36
      DO 35 I=2,ND
      DO 35 J=1,N
   35 D(J,I-1)=D(J,I)
   36 DO 37 J=1,N
   37 D(J,ND)=T(J)
   38 CONTINUE
C
C     UPDATE F AND Z
C
      IF(NA-1) 26,25,23
   23 DO 24 I=NA,2,-1
      Z(I)=Z(I-1)
   24 F(I)=F(I-1)
   25 F(1)=-Y
      Z(1)=-YEST
   26 IF(NB-1) 30,29,27
   27 DO 28 I=NB,2,-1
      Z(NA+I)=Z(NA+I-1)
   28 F(NA+I)=F(NA+I-1)
   29 F(NA+1)=U
      Z(NA+1)=U
   30 CONTINUE
C
      RETURN
      END
```

```
      SUBROUTINE RTMLE(T,V,VTTIN,U,Y,RES,NA,NB,NC,INIT,INEW,IA)
C
C     REAL TIME MAXIMUM LIKELIHOOD ESTIMATION
C     THE FOLLOWING MODEL IS USED
C     A(Q)*Y(T)=B(Q)*U(T)+C(Q)*E(T)
C     A(Q)=1+A(1)*Q**(-1)+...A(NA)*Q**(-NA)
C     B(Q)=  B(1)*Q**(-1)+...B(NB)*Q**(-NB)
C     C(Q)=1+C(1)*Q**(-1)+...C(NC)*Q**(-NC)
C     E(T) WHITE NOISE
C
C     AUTHOR TORSTEN SÖDERSTRÖM, 1973-02-21
C     REFERENCE REPORTS 7308,7312(C)
C
C     THE SUBROUTINE PERFORMS THE MODIFICATION OF THE ESTIMATES
C     FOR ONE SAMPLING INTERVAL AND A NEW CALL MUST BE DONE
C     FOR EVERY NEW SAMPLING EVENT.
C
C     T - VECTOR OF ORDER (NA+NB+NC) CONTAINING THE PARAMETER ESTIMATES
C       T = (A(1),...A(NA),B(1),...B(NB),C(1),...C(NC))
C     V - LOSS FUNCTION
C     VTTIN - MATRIX OF ORDER (NA+NB+NC) * (NA+NB+NC) THE INVERSE OF
C       THE MATRIX OF THE SECOND ORDER DERIVATIVES
C     U - THE LAST INPUT
C     Y - THE LAST OUTPUT
C     RES - THE LAST RESIDUAL (COMPUTED IN RTMLE)
C     NA,NB,NC - NUMBER OF PARAMETERS (MIN 0,MAX 10)
C       THEY CAN BE CHOSEN INDEPENTLY OF EACH OTHER
C       (NA+NB+NC) .GT.0
C     INIT - IF INIT IS 1 V,VTTIN AND THE INTERNAL VARIABLES ARE GIVEN
C       APPROPRIATE INITIAL VALUES
C       ELSE THE ALGORITHM IS APPLIED STRAIGHTFORWARD
C     INEW - IF INEW IS 1 T IS NOT UPDATED
C       ELSE THE ALGORITHM IS APPLIED STRAIGHTFORWARD
C     IA - DIMENSION PARAMETER
C
C     SUBROUTINE REQUIRED
C           NSTAB
C
      DIMENSION T(1),VTTIN(IA,1)
      DIMENSION X(10),Z(30),REST(30),TT(30),TC(10)
C
      N=MAX0(NA,NB,NC)
      NN=NA+NB+NC
      N2=N*2
      N3=N*3
      DO 1 I=1,N3
    1 TT(I)=0.
      IF(NA.EQ.0) GO TO 3
      DO 2 I=1,NA
    2 TT(I)=T(I)
    3 IF(NB.EQ.0) GO TO 5
      DO 4 I=1,NB
    4 TT(I+N)=T(I+NA)
    5 IF(NC.EQ.0) GO TO 7
      DO 6 I=1,NC
    6 TT(I+N2)=T(I+NA+NB)
    7 CONTINUE
C
C     TEST OF INIT
C
      IF(INIT.NE.1) GO TO 20
      V=0.
      UOLD=0.
```

```
      YOLD=0.
      RESOLD=0.
      DO 11 I=1,N
   11 X(I)=0.
      DO 13 I=1,NN
      DO 12 J=1,NN
   12 VTTIN(I,J)=0.
   13 VTTIN(I,I)=100.
      DO 14 I=1,N3
   14 Z(I)=0.
   20 CONTINUE
C
C     COMPUTATION OF RES
C
      F=Y+TT(1)*YOLD-TT(N+1)*UOLD-TT(N2+1)*X(1)
      IF(N.EQ.1) GO TO 30
      F=F+X(2)
      IF(N.EQ.2) GO TO 23
      DO 22 I=3,N
      IM=I-1
   22 X(IM)=X(I)+TT(IM)*YOLD-TT(N+IM)*UOLD-TT(N2+IM)*X(1)
   23 X(N)=TT(N)*YOLD-TT(N2)*UOLD-TT(N3)*X(1)
   30 X(1)=F
      RES=X(1)
C
C     COMPUTATION OF REST, THE GRADIENT OF RES
C
      H1=YOLD
      H2=-UOLD
      H3=-RESOLD
      DO 31 I=1,N
   31 H1=H1-TT(N2+I)*Z(I)
      DO 32 I=1,N
   32 H2=H2-TT(N2+I)*Z(N+I)
      DO 33 I=1,N
   33 H3=H3-TT(N2+I)*Z(I+N2)
C
      IF(N-1) 36,36,34
   34 DO 35 I=N,2,-1
      Z(I)=Z(I-1)
      Z(N+I)=Z(N+I-1)
   35 Z(N2+I)=Z(N2+I-1)
   36 Z(1)=H1
      Z(N+1)=H2
      Z(N2+1)=H3
C
      IF(NA.EQ.0) GO TO 38
      DO 37 I=1,NA
   37 REST(I)=Z(I)
   38 IF(NB.EQ.0) GO TO 40
      DO 39 I=1,NB
   39 REST(NA+I)=Z(N+I)
   40 IF(NC.EQ.0) GO TO 42
      DO 41 I=1,NC
   41 REST(NA+NB+I)=Z(N2+I)
   42 CONTINUE
C
C     COMPUATION OF V AND VTTIN
C     THE HELP VARIABLES TT ARE USED FOR NEW PURPOSES
C
      DO 51 I=1,NN
      TT(I)=0.
      DO 51 J=1,NN
   51 TT(I)=TT(I)+VTTIN(I,J)*REST(J)
```

IV. EXAMPLES

REAL TIME IDENTIFICATION

RTLS IS USED

TRUE PARAMETER VALUES

-.80000	1.00000	.70000	1.00000

NSA	NSB	NSC	NMA	NMB	NMC	M
1	1	1	1	1	0	1000

TIME AND PARAMETER ESTIMATES

25	-.82198	.91886	-.66825
50	-.83561	.83977	1.2158
75	-.83915	.83501	-1.5295
100	-.81303	.84543	.75452
125	-.82491	.80246	.63829
150	-.82548	.76115	1.5123
175	-.81637	.86362	-.50909
200	-.87116	.91681	1.1722
225	-.87128	.88836	-2.3573
250	-.87215	.86199	.31607
275	-.86908	.89381	-3.9147
300	-.87343	.87715	-.14908-01
325	-.87174	.90569	-1.5948
350	-.87540	.92935	.32646
375	-.86997	.91316	2.4994
400	-.86796	.93540	.59500
425	-.86908	.94283	.79663
450	-.87003	.95261	1.0017
475	-.86404	.93257	.85136
500	-.86065	.93956	-1.2451
525	-.86438	.95269	-.71441-01
550	-.86354	.95504	1.8960
575	-.86366	.95250	-1.1308
600	-.86435	.95491	-.70440
625	-.86124	.95768	1.0446
650	-.86006	.96836	-1.9974
675	-.85994	.96233	-1.1091
700	-.85988	.96453	.57874-01
725	-.85824	.95510	-3.2733
750	-.85589	.95456	-.60282
775	-.85545	.94955	-.19807
800	-.85332	.94279	.87696
825	-.85382	.95381	.34205
850	-.85282	.95189	-2.5588
875	-.85419	.95774	.91153-01
900	-.85488	.96395	1.7011
925	-.85577	.96952	-.78278
950	-.85502	.97434	-1.7521
975	-.85432	.97404	-1.7898
1000	-.85678	.97658	-.30294

VARIANCE OF THE MODEL ERROR .15527
VARIANCE OF THE RESIDUALS 1.4434
VARIANCE OF THE OUTPUT 1.3900

TIME FOR THIS IDENTIFICATION WAS 632 MS

```
      A=1.
      DO 52 I=1,NN
   52 A=A+TT(I)*REST(I)
      V=V+RES*RES/A/2.
      DO 53 I=1,NN
      DO 53 J=1,I
      VTTIN(I,J)=VTTIN(I,J)-TT(I)*TT(J)/A
   53 VTTIN(J,I)=VTTIN(I,J)
C
C     UPDATE T
C
      IF(INEW.EQ.1) GO TO 70
      DO 61 I=1,NN
      TT(I)=0.
      DO 61 J=1,NN
   61 TT(I)=TT(I)+VTTIN(I,J)*REST(J)*RES
      DO 62 I=1,NN
   62 T(I)=T(I)-TT(I)
      IF(NC.EQ.0) GO TO 70
C
      S=1.0
   63 DO 64 I=1,NC
   64 TC(I)=T(NA+NB+I)
      IF(NSTAB(TC,NC).EQ.0) GO TO 70
      S=S/2.
      DO 65 I=1,NN
   65 T(I)=T(I)+S*TT(I)
      GO TO 63
   70 CONTINUE
C
C     UPDATE INTERNAL VARIABLES
C
      UOLD=U
      YOLD=Y
      RESOLD=RES
C
      RETURN
      END
```

REAL TIME IDENTIFICATION

RTGLS IS USED

TRUE PARAMETER VALUES

-.10000	-.56000	1.00000	.70000	1.00000

NSA	NSB	NSC	NMA	NMB	NMC	M
2	2	0	1	1	1	1000
20						

TIME AND PARAMETER ESTIMATES

25	-.54160	.83134	.58728	-1.5533
50	-.72534	.82923	.48452	-.51553
75	-.73698	.90437	.51856	-.28179
100	-.73979	.82698	.56153	.64985
125	-.75386	.85182	.61871	-.31014
150	-.77275	.81743	.61110	1.8951
175	-.75675	.88406	.67211	.21438
200	-.78684	.92771	.65551	1.9513
225	-.79078	.91045	.64353	-.76754
250	-.79298	.90522	.62905	-.72638
275	-.79235	.92595	.61952	-1.6629
300	-.79922	.93366	.66619	1.6801
325	-.79553	.93013	.73300	-.49260
350	-.79873	.95136	.72332	.87552
375	-.79463	.93720	.72108	1.9774
400	-.79369	.94593	.71928	.34695
425	-.78886	.95547	.72051	.39254
450	-.79115	.96314	.72263	.68217
475	-.79253	.93807	.71349	.78275
500	-.79071	.93286	.70710	-.41999
525	-.79733	.94237	.71064	.62415
550	-.79487	.94791	.70531	1.3827
575	-.79552	.95074	.70677	-1.5854
600	-.79502	.95928	.70301	-.83563
625	-.79386	.95549	.70246	.96171
650	-.79466	.96351	.70215	-1.8012
675	-.79817	.95638	.70298	-.68744
700	-.79824	.96041	.70227	-.56621
725	-.79553	.96070	.70007	-2.6319
750	-.79312	.96177	.70566	-.18469
775	-.79126	.96217	.71021	.52590
800	-.79154	.96068	.71290	.96249
825	-.79215	.96598	.71242	-.18360
850	-.79142	.96607	.71343	-1.7410
875	-.79160	.96844	.71983	.60176
900	-.79351	.97699	.71433	1.2120
925	-.79414	.98195	.71270	.55895
950	-.79254	.98780	.71244	-1.5919
975	-.79128	.98772	.70877	-1.1199
1000	-.79210	.99740	.70724	-.35692

VARIANCE OF THE MODEL ERROR .23297-02
VARIANCE OF THE RESIDUALS 1.0008
VARIANCE OF THE OUTPUT 1.0002

TIME FOR THIS IDENTIFICATION WAS 1390 MS

REAL TIME IDENTIFICATION

RTIVM1 IS USED

TRUE PARAMETER VALUES

-.80000	1.00000	.70000	1.00000

NSA	NSB	NSC	NMA	NMB	NMC	M
1	1	1	1	1	0	1000

TIME AND PARAMETER ESTIMATES

25	-.95544	.87644	-.55289
50	-.77277	.85547	1.0866
75	-.85522	.83321	-1.6102
100	-.73579	.84838	.73633
125	-.70406	.82290	.14895
150	-.65761	.79971	1.6148
175	-.70189	.89337	-.47183
200	-.75350	.95078	.75360
225	-.73799	.91387	-2.1104
250	-.71838	.87656	.37591
275	-.73959	.89890	-3.3109
300	-.75245	.88108	-.30459
325	-.73940	.90990	-2.1534
350	-.76611	.92761	-.63934-01
375	-.73426	.91490	2.5676
400	-.74306	.93352	.88526
425	-.74957	.93955	.53630-01
450	-.75030	.94523	1.1627
475	-.71450	.92437	.76017
500	-.70108	.93136	-.70149
525	-.71898	.94573	.24682
550	-.72677	.94626	1.8514
575	-.72252	.94561	-1.6638
600	-.73999	.94383	-.49981
625	-.73063	.94586	.67052
650	-.74278	.96021	-1.9463
675	-.73776	.95873	-.61694
700	-.74213	.95819	.70299-01
725	-.75085	.94899	-3.3499
750	-.74816	.94978	-.56659
775	-.74752	.94378	.53131-01
800	-.74908	.93880	1.0089
825	-.75721	.94959	.36169
850	-.75677	.94714	-2.5757
875	-.75819	.95277	-.32847
900	-.77346	.96175	1.8432
925	-.78039	.97021	-.67850
950	-.79005	.97469	-1.8139
975	-.78934	.97391	-2.0426
1000	-.80321	.97569	-.51048-01

VARIANCE OF THE MODEL ERROR .10479-02
VARIANCE OF THE RESIDUALS 1.4862
VARIANCE OF THE OUTPUT 1.4583

TIME FOR THIS IDENTIFICATION WAS 608 MS

REAL TIME IDENTIFICATION

RTIVM2 IS USED

TRUE PARAMETER VALUES

-.80000 1.00000 .70000 1.00000

NSA	NSB	NSC	NMA	NMB	NMC	M
1	1	1	1	1	0	1000
0						

TIME AND PARAMETER ESTIMATES

25	-.78143	.93199	-.71347
50	-.78427	.85232	1.1523
75	-.79620	.84041	-1.4283
100	-.76435	.84719	.74140
125	-.76912	.81184	.41431
150	-.74443	.77972	1.5653
175	-.73793	.88394	-.48417
200	-.81247	.93371	.96987
225	-.80764	.90050	-2.2364
250	-.80326	.86849	.34308
275	-.79184	.89681	-3.5480
300	-.79761	.87958	-.19454
325	-.79650	.90806	-1.8970
350	-.80047	.92813	.58992-01
375	-.79114	.91414	2.5407
400	-.78917	.93419	.77560
425	-.79181	.94068	.31573
450	-.79168	.94776	1.1049
475	-.78379	.92815	.80188
500	-.77844	.93532	-.96861
525	-.78673	.94896	.98055-01
550	-.78443	.94994	1.8696
575	-.78385	.94858	-1.4421
600	-.78356	.94769	-.57052
625	-.77958	.95027	.80952
650	-.78067	.96282	-1.9619
675	-.78326	.96005	-.79846
700	-.78364	.96041	.65910-01
725	-.78053	.95066	-3.3265
750	-.77803	.95109	-.57644
775	-.77532	.94525	-.11400-01
800	-.77308	.93970	.97734
825	-.77400	.95031	.35834
850	-.77262	.94790	-2.5733
875	-.77578	.95366	-.25076
900	-.77931	.96189	1.8298
925	-.78264	.97017	-.68094
950	-.78242	.97471	-1.8193
975	-.77855	.97387	-2.0784
1000	-.78254	.97532	.45595-01

VARIANCE OF THE MODEL ERROR .15722-01
VARIANCE OF THE RESIDUALS 1.5181
VARIANCE OF THE OUTPUT 1.4867

TIME FOR THIS IDENTIFICATION WAS 603 MS

REAL TIME IDENTIFICATION

RTIVM2 IS USED

TRUE PARAMETER VALUES

-.80000   1.00000   .70000   1.00000

i	NSA	NSB	NSC	NMA	NMB	NMC	M
3	1	1	1	1		0	1000

TIME AND PARAMETER ESTIMATES

25	-.79620	.92720	-.65261
50	-.69410	.87437	.94596
75	.33986	.98333	2.1329
100	-.49359	.85698	.66352
125	-1.3550	.71335	3.2287
150	-2.1619	.45492	.48497
175	-.81961	.86279	-.51465
200	-1.3202	.78750	2.8414
225	-1.5715	.75471	-3.4981
250	-1.3879	.81334	.11232
275	-1.2439	.87924	-5.4036
300	-7.8700	.65334	204.36
325	-1.5981	.88282	1.5106
350	2.9415	.86683	-7.8803
375	-2.3839	.89434	1.6075
400	-2.0885	.95411	-1.8211
425	-2.1479	.97835	12.414
450	-.24694	.91399	1.9248
475	-.54102	.91479	.64311
500	-.76729	.93474	-.97680
525	-.10886	.91634	1.7299
550	1.0634	.83091	1.2173
575	10.091	.41551	-18.719
600	-2.7073	1.1194	-3.3839
625	-2.7418	1.1283	7.9612
650	-1.2456	.99523	-2.1552
675	1.4679	.89324	14.525
700	.15888-01	.91718	.15075
725	-.26578	.92128	-3.6243
750	.10962	.91145	-.27893
775	.19419-01	.90256	2.0012
800	-.53766	.93065	1.2981
825	-.72668	.94823	.36893
850	-.71509	.94505	-2.5785
875	-.35473	.93174	-1.9312
900	-2.1244	.99853	-.43261
925	-.48484	.97283	-.27062
950	-.59242	.97569	-1.9718
975	-.38666	.97298	-3.3732
1000	-.59052	.97208	1.0043

VARIANCE OF THE MODEL ERROR .54364
VARIANCE OF THE RESIDUALS 2.2241
VARIANCE OF THE OUTPUT 1.8272

TIME FOR THIS IDENTIFICATION WAS 728 MS

REAL TIME IDENTIFICATION

RTMLE IS USED

TRUE PARAMETER VALUES

-.80000 1.00000 .70000 1.00000

NSA NSB NSC NMA NMB NMC M

1 1 1 1 1 1 1000

200 50

TIME AND PARAMETER ESTIMATES

25	-.78464	.76321	.28340	-.82286
50	-.75708	.76722	.48573	.12174
75	-.75920	.76905	.50399	-.98879
100	-.70290	.77852	.57390	.88443
125	-.71253	.83919	.55704	.12407
150	-.71255	.84102	.60503	1.8253
175	-.71508	.90244	.57693	.66918-01
	W AND VTEST	5.9582	1.0300	
200	-.79485	.95529	.57172	1.5773
225	-.79485	.95529	.57172	-1.2964
250	-.79691	.95496	.56488	-.34026
275	-.77120	.94102	.68346	-2.1987
300	-.80024	.94907	.59239	1.4293
325	-.80530	1.0157	.51002	-.82821
350	-.82101	1.0334	.53163	.97714
375	-.79218	1.0216	.57287	1.9833
400	-.78562	1.0325	.58506	.41246
425	-.78709	1.0405	.59351	.44801
	W AND VTEST	1.0273	1.0300	
450	-.79142	1.0677	.60207	.76668
475	-.76235	1.0452	.63972	.90552
500	-.75425	1.0372	.65719	-.34071
525	-.77077	1.0497	.64668	.66879
550	-.76825	1.0426	.65920	1.5138
575	-.76873	1.0528	.66034	-1.4899
600	-.77014	1.0333	.66754	-.74995
625	-.76006	1.0393	.67808	.81905
650	-.76191	1.0517	.68182	-1.8752
675	-.76267	1.0358	.67815	-.67012
700	-.76281	1.0273	.67917	-.49524
725	-.75709	1.0104	.68523	-2.7279
750	-.75368	1.0086	.67862	-.19582
775	-.75411	.99765	.67094	.55389
800	-.75094	.99297	.66805	.99550
825	-.75299	.99476	.66847	-.14924
850	-.74978	.98876	.66839	-1.7895
875	-.75362	1.0014	.65956	.42299
900	-.75940	1.0041	.66603	1.3350
925	-.76505	1.0058	.66678	.51812
950	-.76492	1.0023	.66682	-1.6765
975	-.76370	.99702	.67360	-1.2620
1000	-.76891	.99949	.67455	-.33847

VARIANCE OF THE MODEL ERROR .27723-01
VARIANCE OF THE RESIDUALS 1.0064
VARIANCE OF THE OUTPUT 1.0053

TIME FOR THIS IDENTIFICATION WAS 1318 MS